

Applications des bases de Gröbner à la géométrie élémen- taire

Sommaire

1.	Traduction algébrique de problèmes de géométrie	115
2.	Conséquences d'un système d'équations algébriques et radical d'un idéal	118
3.	Hypothèses implicites de genericité dans les théorèmes de géométrie	120
4.	Découvrir ou redécouvrir des théorèmes de géométrie	122

Dans cette partie on illustre l'utilisation des bases de Gröbner pour retrouver de manière effective des théorèmes de géométrie élémentaire.

§ 1 Traduction algébrique de problèmes de géométrie

VIII.1.1. Droite de Gauss-Newton.— On considère un quadrilatère convexe $ABCD$ sans côtés parallèles et on note E le point d'intersection de (AB) avec (CD) et F le point d'intersection de (AD) avec (BC) . Soient I, J et K les milieux respectifs des trois diagonales $[AC]$, $[BD]$ et $[EF]$ de ce *quadrilatère complet*. Alors les trois points I, J et K sont alignés.




FIGURE VIII.1.: Droite de Gauss-Newton

On peut retrouver ce résultat en traduisant les hypothèses géométriques par des relations polynomiales : pour cela, on considère les coordonnées des points A, B, C, D, F, I, J, K dans la plan euclidien. Sans perdre de généralités, on peut supposer que A est à l'origine et B sur l'axe des abscisses. On obtient ainsi

$$\begin{aligned} A &= (0, 0), & B &= (x_0, 0), & C &= (x_1, x_2), & D &= (x_3, x_4), & E &= (x_5, 0), \\ F &= (x_6, x_7), & I &= (x_8, x_9), & J &= (x_{10}, x_{11}), & K &= (x_{12}, x_{13}). \end{aligned}$$

On traduit les hypothèses géométriques en terme d'équations polynomiales :

– $E \in (AB) \cap (CD)$ se traduit par $\begin{vmatrix} x_5 - x_1 & x_3 - x_1 \\ -x_2 & x_4 - x_2 \end{vmatrix} = 0$, c'est-à-dire par

$$-x_1x_4 + x_2x_3 - x_2x_5 + x_4x_5 = 0$$

(ici comme on a pris A, B et E sur l'axe des abscisses, l'hypothèse $E \in (AB)$ est déjà vérifiée).

– $F \in (AD) \cap (BC)$ se traduit par $\begin{vmatrix} x_6 & x_3 \\ x_7 & x_4 \end{vmatrix} = 0$ et $\begin{vmatrix} x_6 - x_0 & x_1 - x_0 \\ -x_7 & x_2 \end{vmatrix} = 0$, c'est-à-dire par

$$-x_3x_7 + x_4x_6 = 0 \text{ et } -x_0x_2 + x_0x_7 - x_1x_7 + x_2x_6 = 0.$$

– I est le milieu de $[AC]$ se traduit par

$$2x_8 - x_1 = 0 \text{ et } 2x_9 - x_2 = 0.$$

– J est le milieu de $[BD]$ se traduit par

$$2x_{10} - x_0 - x_3 = 0 \text{ et } 2x_{11} - x_4 = 0.$$

– K est le milieu de $[EF]$ se traduit par

$$2x_{12} - x_5 - x_6 = 0 \text{ et } 2x_{13} - x_7 = 0.$$

La conclusion, les trois points I, J et K sont alignés, est équivalente à

$$\begin{vmatrix} x_{10} - x_8 & x_{12} - x_8 \\ x_{11} - x_9 & x_{13} - x_9 \end{vmatrix} = 0 \text{ c'est-à-dire à}$$

$$x_8x_{11} - x_8x_{13} - x_9x_{10} + x_9x_{12} + x_{10}x_{13} - x_{11}x_{12} = 0.$$

On se place dans l'anneau $\mathbb{R}[x_0, x_1, \dots, x_{13}]$ et on pose

$$\begin{aligned} f_1 &= -x_1x_4 + x_2x_3 - x_2x_5 + x_4x_5, & f_2 &= -x_3x_7 + x_4x_6, \\ f_3 &= -x_0x_2 + x_0x_7 - x_1x_7 + x_2x_6, & f_4 &= 2x_8 - x_1, \\ f_5 &= 2x_9 - x_2, & f_6 &= 2x_{10} - x_0 - x_3, \\ f_7 &= 2x_{11} - x_4 = 0, & f_8 &= 2x_{12} - x_5 - x_6, \\ f_9 &= 2x_{13} - x_7, & f &= x_8x_{11} - x_8x_{13} - x_9x_{10} + x_9x_{12} + x_{10}x_{13} - x_{11}x_{12}. \end{aligned}$$

On peut alors vérifier en calculant une base de Gröbner (ou directement avec Sage) que $f \in \langle f_1, f_2, f_3, f_4, f_5, f_6, f_7, f_8, f_9 \rangle$.

On en déduit que $f = 0$ est conséquence du système $f_1 = f_2 = f_3 = f_4 = f_5 = f_6 = f_7 = f_8 = f_9 = 0$.

Le code Sage pour vérifier que $f \in \langle f_1, f_2, f_3, f_4, f_5, f_6, f_7, f_8, f_9 \rangle$:

```
sage: R= PolynomialRing(QQ, 'x', 14)
sage: x= R.gens()
sage: f1 = (x[4]-x[2])*(x[5]-x[1])+x[2]*(x[3]-x[1])
sage: f2 = x[4]*x[6]-x[3]*x[7]
sage: f3 = x[2]*(x[6]-x[0])-(x[1]-x[0])*x[7]
sage: f4 = 2*x[8]-x[1]
sage: f5 = 2*x[9]-x[2]
sage: f6 = 2*x[10] - x[3]-x[0]
sage: f7 = 2*x[11] - x[4]
sage: f8 = 2*x[12] - x[5]-x[6]
sage: f9 = 2*x[13] -x[7]
sage: f = (x[10]-x[8])*(x[13]-x[9]) - (x[11]-x[9])*(x[12]-x[8])
sage: I = (f1, f2, f3, f4, f5, f6, f7, f8, f9)*R
sage: f in I
True
```

En utilisant l'ordre lexicographique et le théorème d'élimination, on peut retrouver directement la propriété de colinéarité des points I, J, K :

```
sage: R= PolynomialRing(QQ, 'x', 14, order='lex')
sage: x= R.gens()
sage: f1 = (x[4]-x[2])*(x[5]-x[1])+x[2]*(x[3]-x[1])
sage: f2 = x[4]*x[6]-x[3]*x[7]
sage: f3 = x[2]*(x[6]-x[0])-(x[1]-x[0])*x[7]
sage: f4 = 2*x[8]-x[1]
sage: f5 = 2*x[9]-x[2]
sage: f6 = 2*x[10] - x[3]-x[0]
sage: f7 = 2*x[11] - x[4]
sage: f8 = 2*x[12] - x[5]-x[6]
sage: f9 = 2*x[13] -x[7]
sage: I = (f1, f2, f3, f4, f5, f6, f7, f8, f9)*R
sage: G = I.groebner_basis()
sage: print G
[x0 + x3 - 2*x10, x1 - 2*x8, x2 - 2*x9, x3*x9 + x6*x9 - x6*x11 -
2*x8*x13 - 2*x9*x10 + 2*x10*x13, x3*x13 - x6*x11, x4 - 2*x11, x5 + x6 -
2*x12, x6*x8*x9*x13 - 1/2*x6*x8*x13^2 + 1/2*x6*x9^2*x10 -
1/2*x6*x9^2*x12 - x6*x9*x10*x13 + 1/2*x6*x10*x13^2 - x8^2*x13^2 -
x8*x9*x10*x13 + x8*x10*x13^2 + x8*x12*x13^2 + x9*x10*x12*x13 -
x10*x12*x13^2, x6*x9*x11 + x6*x9*x13 - x6*x11*x13 - 2*x8*x13^2 -
2*x9*x10*x13 + 2*x10*x13^2, x7 - 2*x13, x8*x11 - x8*x13 - x9*x10 +
x9*x12 + x10*x13 - x11*x12]
```

VIII.1.2. Remarque.— Pour les calculs avec Sage ci-dessus, on se limite à l'anneau des polynômes à coefficients rationnels, car les hypothèses et la conclusion sont à coefficients rationnels et tous calculs intermédiaires (divisions, réductions, calcul de bases de Gröbner) ne font également intervenir que des des polynômes à coefficients rationnels.

Exercice 1. — En traduisant les hypothèses géométriques dans le repère affine $(A, \overrightarrow{AB}, \overrightarrow{AD})$, retrouver le résultat sur la droite de Gauss-Newton.

VIII.1 Proposition. — Soit A, B, C, D, E, F six points distincts du plan euclidien orienté. Les énoncés géométriques suivants s'expriment par des équations polynomiales :

1. les droites (AB) et (CD) sont parallèles ;
2. les droites (AB) et (CD) sont perpendiculaires ;
3. les points A, B et C sont alignés ;
4. les distances AB et CD sont égales ;
5. le point C appartient au cercle de centre A et de rayon AB ;
6. le point C est le milieu du segment $[AB]$;
7. les angles orientés $(\overrightarrow{AB}, \overrightarrow{AC})$ et $(\overrightarrow{DE}, \overrightarrow{DF})$ sont égaux modulo π ;
8. la droite (BD) est la bissectrice de l'angle \widehat{ABC} ;
9. le point C appartient à la médiatrice de $[AB]$.

Exercice 2. — Montrer la proposition VIII.1. Pour le point 7 rappelons que

$$\det(\overrightarrow{AB}, \overrightarrow{AC}) = AB \times AC \times \sin(\widehat{ABC}) \text{ et } \overrightarrow{AB} \cdot \overrightarrow{AC} = AB \times AC \times \cos(\widehat{ABC})$$

Exercice 3. — Soit ABC un triangle. Traduire en termes d'équations polynomiales les théorèmes de géométrie suivants :

1. Les trois hauteurs de ABC se coupent en unique point, appelé l'orthocentre de ABC .
2. Les trois médianes de ABC se coupent en unique point, appelé le centre de gravité de ABC .
3. Le centre du cercle circonscrit, l'orthocentre et le centre gravité du triangle ABC sont alignés. (La droite contenant ces trois points est appelée droite d'Euler.)

§ 2 Conséquences d'un système d'équations algébriques et radical d'un idéal

VIII.2.1. Conséquences d'un système d'équations algébriques.— Dans la partie précédente, on a étudié un théorème de géométrie dont les hypothèses et les conclusions

peuvent être traduites en termes de systèmes d'équations polynomiales. De manière abstraite, la question est de résoudre le problème suivant :

Étant donnés des polynômes f_1, \dots, f_s, g de $\mathbb{K}[x_1, \dots, x_n]$, est-ce que

$$g(x_1, \dots, x_n) = 0 \text{ est conséquence du système } \left\{ \begin{array}{l} f_1(x_1, \dots, x_n) = 0 \\ \vdots \\ f_s(x_1, \dots, x_n) = 0 \end{array} \right. ?$$

Définition VIII.2. — Soit f_1, \dots, f_s, g des polynômes de $\mathbb{K}[x_1, \dots, x_n]$. On dit que l'équation $g = 0$ est une *conséquence* du système d'équations $f_1 = \dots = f_s = 0$ si $g(a_1, \dots, a_n) = 0$ pour tout $(a_1, \dots, a_n) \in \mathbf{V}(f_1, \dots, f_s)$, autrement dit si

$$g \in I(\mathbf{V}(f_1, \dots, f_s)).$$

VIII.2.2. Radical d'un idéal.—

VIII.3 Proposition. — Soit I un idéal de $\mathbb{K}[x_1, \dots, x_n]$. L'ensemble \sqrt{I} suivant, appelé radical de I ,

$$\sqrt{I} := \{f \in \mathbb{K}[x_1, \dots, x_n] \mid f^m \in I \text{ pour un entier } l\}$$

est un idéal contenant I .

Exercice 4. — 1. Montrer la proposition précédente.

2. Donner un exemple d'idéal strictement inclus dans son radical.

VIII.4 Proposition. — Soit f_1, \dots, f_s, g des polynômes de $\mathbb{K}[x_1, \dots, x_n]$.

Si $g \in \sqrt{\langle f_1, \dots, f_s \rangle}$ alors l'équation $g = 0$ est une conséquence système d'équations $f_1 = \dots = f_s = 0$.

Exercice 5. — 1. Montrer la proposition précédente.

2. Donner un exemple de deux polynômes f et g de $\mathbb{R}[x, y]$ tel que $g = 0$ est conséquence de $f = 0$ mais $g \notin \sqrt{\langle f \rangle}$.

Dans le cas du corps \mathbb{C} (et plus généralement d'un corps algébriquement clos), les deux notions sont équivalentes par le théorème des zéros de Hilbert :

VIII.5 Théorème (admis !). — Soit $f_1, \dots, f_s \in \mathbb{C}[x_1, \dots, x_n]$ alors

$$I(\mathbf{V}(f_1, \dots, f_s)) = \sqrt{\langle f_1, \dots, f_s \rangle}.$$

Autrement dit, pour tout polynôme $g \in \mathbb{C}[x_1, \dots, x_n]$, l'équation $g = 0$ est conséquence du système $f_1 = \dots = f_s = 0$ si et seulement si $g \in \sqrt{\langle f_1, \dots, f_s \rangle}$.

Exercice 6. — Soit $f_1, \dots, f_s \in \mathbb{R}[x_1, \dots, x_n]$. On note $\mathbf{V}_{\mathbb{C}}(f_1, \dots, f_s)$ l'ensemble algébrique affine formé des points de \mathbb{C}^n satisfaisant le système $f_1 = \dots = f_s = 0$. Soit $g \in \mathbb{R}[x_1, \dots, x_n]$. Montrer que

$$g \in \sqrt{\langle f_1, \dots, f_s \rangle} \subseteq \mathbb{R}[x_1, \dots, x_n]$$

si et seulement si

$$g \in I(\mathbf{V}_{\mathbb{C}}(f_1, \dots, f_s)) \subseteq \mathbb{C}[x_1, \dots, x_n].$$

VIII.2.3. Problème de l'appartenance au radical d'un idéal. — La théorème suivant donne un critère pour déterminer si un polynôme appartient au radical d'un idéal :

VIII.6 Théorème (admis !). — Soit $f_1, \dots, f_s, g \in \mathbb{K}[x_1, \dots, x_n]$. Alors $g \in \sqrt{\langle f_1, \dots, f_s \rangle}$ si et seulement si

$$1 \in \langle f_1, \dots, f_s, 1 - yg \rangle \subseteq \mathbb{K}[x_1, \dots, x_n, y].$$


On peut donc répondre à ce problème en calculant une base de Gröbner de l'idéal $\langle f_1, \dots, f_s, 1 - yg \rangle$.

Exercice 7. — Soit $f_1, \dots, f_s, g \in \mathbb{K}[x_1, \dots, x_n]$ et $I = \langle f_1, \dots, f_s, 1 - yg \rangle \subseteq \mathbb{K}[x_1, \dots, x_n, y]$. Soit G une base de Gröbner de I pour un ordre monomial fixé.

Montrer que $g \in \sqrt{\langle f_1, \dots, f_s \rangle}$ si et seulement si G contient un polynôme constant.

§ 3 Hypothèses implicites de genericité dans les théorèmes de géométrie

VIII.3.1. Exemple. — Soit A, B et C trois points du plan affine. Soit D le milieu de $[AB]$ et E le milieu de $[CD]$. Si on note F le point d'intersection des droites (AE) et (BC) , on a $FB = 2FC$.



Sans perte de généralité, on peut supposer que D est l'origine du plan et A et B sont sur l'axe des abscisses. Ainsi, on peut choisir les coordonnées suivantes pour les points de ce problème :

$$A = (-x_0, 0), \quad B = (x_0, 0), \quad C = (x_1, x_2), \quad D = (0, 0), \quad E = (x_3, x_4), \quad F = (x_5, x_6).$$

Ce choix particulier de coordonnées assure que D est le milieu de $[AB]$. Traduisons en termes algébriques les autres hypothèses :

– E est le milieu de $[CD]$ se traduit par

$$2x_3 - x_1 = 0 \text{ et } 2x_4 - x_2 = 0.$$

– F est le point d'intersection des droites (AE) et (BC) se traduit par $\begin{vmatrix} x_5 + x_0 & x_3 + x_0 \\ x_6 & x_4 \end{vmatrix} =$

$$0 \text{ et } \begin{vmatrix} x_5 - x_0 & x_1 - x_0 \\ x_6 & x_2 \end{vmatrix} = 0, \text{ c'est-à-dire par}$$

$$x_0x_4 - x_0x_6 - x_3x_6 + x_4x_5 = 0 \text{ et } -x_0x_2 + x_0x_6 - x_1x_6 + x_2x_5 = 0.$$

Comme $F \in [BC]$, la conclusion $FB = 2FC$ est équivalente à

$$3x_5 - 2x_1 - x_0 = 0 \text{ et } 3x_6 - 2x_2 = 0.$$

Considérons maintenant les polynômes de $\mathbb{R}[x_0, x_1, x_2, x_3, x_4, x_5, x_6]$ suivants : $f_1 = 2x_3 - x_1$, $f_2 = 2x_4 - x_2$, $f_3 = x_0x_4 - x_0x_6 - x_3x_6 + x_4x_5$, $f_4 = -x_0x_2 + x_0x_6 - x_1x_6 + x_2x_5$, $g_1 = 3x_5 - 2x_1 - x_0$ et $g_2 = 3x_6 - 2x_2$.

Soit $I = \langle f_1, f_2, f_3, f_4 \rangle$. Alors $g_1 \notin I$ et $g_2 \notin I$:

```
sage: R.<x0,x1,x2,x3,x4,x5,x6>= PolynomialRing(QQ, 'x0,x1,x2,x3,x4,x5,x6')
sage: f1 = 2*x3-x1
sage: f2 = 2*x4-x2
sage: f3 = (x5+x0)*x4 -(x3+x0)*x6
sage: f4 = (x5-x0)*x2 -(x1-x0)*x6
sage: g1 = 3*x5-2*x1-x0
sage: g2 = 3*x6-2*x2
sage: I = (f1,f2,f3,f4)*R
sage: g1 in I, g2 in I
(False, False)
```

Mais, en fait g_1 et g_2 n'appartiennent pas non plus au radical \sqrt{I} :

```
sage: R.<x0,x1,x2,x3,x4,x5,x6,y>= PolynomialRing(QQ, 'x0,x1,x2,x3,x4,x5,x6,y')
sage: f1 = 2*x3-x1
sage: f2 = 2*x4-x2
sage: f3 = (x5+x0)*x4 -(x3+x0)*x6
sage: f4 = (x5-x0)*x2 -(x1-x0)*x6
sage: g1 = 3*x5-2*x1-x0
```

```
sage: g2 = 3*x6-2*x2
sage: h1 = 1-y*g1
sage: h2 = 1-y*g2
sage: I1 = (f1, f2, f3, f4, h1)*R
sage: I2 = (f1, f2, f3, f4, h2)*R
sage: 1 in I1, 1 in I2
(False, False)
```

Cela signifie que les équations $g_1 = 0$ et $g_2 = 0$ ne sont pas conséquences du système $f_1 = f_2 = f_3 = f_4 = 0$ au-dessus du corps des complexes. Dans la traduction ci-dessus on omet de traduire l'hypothèse implicite de généralité : les points A , B et C ne sont pas colinéaires ; c'est-à-dire l'hypothèse $x_0x_2 \neq 0$.

Pour inclure l'inéquation $x_0x_2 \neq 0$ parmi les hypothèses, on considère une nouvelle variable t et l'équation $1 - x_0x_2t = 0$. Posons $f_5 = 1 - x_0x_2t$ et $J = \langle f_1, f_2, f_3, f_4, f_5 \rangle \subseteq \mathbb{R}[x_0, x_1, x_2, x_3, x_4, x_5, x_6, t]$. Alors, on vérifie que g_1 et g_2 appartiennent à J :


```
sage: R.<t, x0, x1, x2, x3, x4, x5, x6>= PolynomialRing(RR, 't, x0, x1, x2, x3, x4, x5, x6')
sage: f1 = 2*x3-x1
sage: f2 = 2*x4-x2
sage: f3 = (x5+x0)*x4 - (x3+x0)*x6
sage: f4 = (x5-x0)*x2 - (x1-x0)*x6
sage: f5 = 1-t*x0*x2
sage: g1 = 3*x5-2*x1-x0
sage: g2 = 3*x6-2*x2
sage: I = (f1, f2, f3, f4, f5)*R
sage: g1 in I, g2 in I
(True, True)
```

Exercice 8. — Dans cet exemple, il n'est pas question d'angles. On peut donc considérer le repère affine $(D, \overrightarrow{DB}, \overrightarrow{DC})$. Retrouver ainsi le résultat.

Exercice 9. — Reprendre l'exercice 3 et vérifier à l'aide du logiciel Sage que les conclusions sont des conséquences des hypothèses après ajout éventuel d'hypothèses de généralité.

§ 4 Découvrir ou redécouvrir des théorèmes de géométrie

VIII.4.1. Théorème de Poncelet. — Dans cette partie, nous retrouvons en utilisant les bases de Gröbner, un résultat de Poncelet qui établit une relation entre les rayons des cercles inscrits et circonscrits d'un triangle et la distance entre les deux centres de ces cercles. Soit ABC un triangle, O_1 le centre du cercle circonscrit de rayon r_1 et O_2 le centre du cercle inscrit de rayon r_2 . Notons d la distance O_1O_2 .



Sans perte de généralité, on peut supposer que A et B sont sur l'axe des abscisses et O_2 sur l'axe des ordonnées. On obtient ainsi pour coordonnées :

$$A(x_1, 0), \quad B(x_2, 0) \quad O_2(0, r_2) \quad C(x_3, x_4) \quad O_1(x_5, x_6).$$

L'hypothèse " (AO_2) est la bissectrice de $(\widehat{AB}, \widehat{AC})$ " s'exprime par l'équation

$$\left| \begin{array}{cc} \overrightarrow{AB} \cdot \overrightarrow{AO_2} & \overrightarrow{AC} \cdot \overrightarrow{AO_2} \\ -\det(\overrightarrow{AB}, \overrightarrow{AO_2}) & \det(\overrightarrow{AC}, \overrightarrow{AO_2}) \end{array} \right| = 0.$$

Après calculs, on obtient

$$(x_2 - x_1)((r_2^2 - x_1^2)x_4 - 2x_1r_2(x_3 - x_1)) = 0.$$

De la même façon, on obtient pour l'hypothèse " (BO_2) est la bissectrice de $(\widehat{BA}, \widehat{BC})$ " l'équation

$$(x_2 - x_1)((r_2^2 - x_2^2)x_4 - 2x_2r_2(x_3 - x_2)) = 0.$$

Le fait que O_1 soit à l'intersection des médiatrices des segments $[AB]$ et $[BC]$ correspond aux égalités suivantes

$$\overrightarrow{AB} \cdot \overrightarrow{AO_1} = \overrightarrow{BA} \cdot \overrightarrow{BO_1} \quad \text{et} \quad \overrightarrow{BC} \cdot \overrightarrow{BO_1} = \overrightarrow{CB} \cdot \overrightarrow{CO_1}.$$

Après calculs, on obtient

$$(x_2 - x_1)(2x_5 - x_2 - x_1) = 0 \quad \text{et} \quad 2x_4x_6 + 2x_3x_5 - 2x_2x_5 - x_4^2 - x_3^2 + x_2^2 = 0.$$

Pour d, r_1 , on obtient les équations suivantes :

$$\begin{aligned}d^2 - (x_6 - r_2)^2 - x_5^2 &= 0, \\ r_1^2 - x_6^2 - (x_5 - x_1)^2 &= 0\end{aligned}$$

Comme on ne considère pas le cas dégénéré où $A = B$, on suppose que $x_2 - x_1 \neq 0$ et on pose alors

$$\begin{aligned}f_1 &= (r_2^2 - x_1^2)x_4 - 2x_1r_2(x_3 - x_1), \\ f_2 &= (r_2^2 - x_2^2)x_4 - 2x_2r_2(x_3 - x_2), \\ f_3 &= 2x_5 - x_2 - x_1, \\ f_4 &= 2x_4x_6 + 2x_3x_5 - 2x_2x_5 - x_4^2 - x_3^2 + x_2^2, \\ f_5 &= d^2 - (x_6 - r_2)^2 - x_5^2, \\ f_6 &= r_1^2 - x_6^2 - (x_5 - x_1)^2.\end{aligned}$$

Soit l'idéal $I = \langle f_1, \dots, f_6 \rangle \subseteq \mathbb{R}[d, r_1, r_2, x_1, \dots, x_6]$. À l'aide de Sage(), on calcule une base de Gröbner correspondant à l'ordre lexicographique (inverse) associé à l'ordre alphabétique $d < r_1 < r_2 < x_1 < \dots < x_6$. Cette base comporte 97 polynômes et l'idéal d'élimination $I \cap \mathbb{R}[d, r_1, r_2]$ est trivial d'après le code Sage() ci-dessous

```
sage: R.<d,r1,r2,x1,x2,x3,x4,x5,x6> =
PolynomialRing(QQ,'d,r1,r2,x1,x2,x3,x4,x5,x6',order='invlex')
sage: f1 = (r2^2-x1^2)*x4-2*x1*r2*(x3-x1)
sage: f2 = (r2^2-x2^2)*x4-2*x2*r2*(x3-x2)
sage: f3 = 2*x5-x2-x1
sage: f4 = 2*x4*x6+2*x3*x5-2*x2*x5-x4^2-x3^2+x2^2
sage: f5 = d^2-(x6-r2)^2-x5^2
sage: f6 = r1^2-x6^2-(x5-x1)^2
sage: I = (f1,f2,f3,f4,f5,f6)*R
sage: G = I.groebner_basis()
Polynomial Sequence with 97 Polynomials in 9 Variables
sage: G[96].factor()
(1/4) * r2 * x1 * (-2*r1*r2 - r1^2 + d^2) * (-2*r1*r2 + r1^2 - d^2) *
(-x1^2 - r2^2 - 2*r1*r2 - r1^2 + d^2) * (-x1^2 - r2^2 + 2*r1*r2 - r1^2 +
d^2) * (r2^2*x1^2 + r2^4 - 2*r1^2*r2^2 - 2*d^2*r2^2 + r1^4 - 2*d^2*r1^2
+ d^4)
```

On ne trouve pas ainsi une relation entre r_1, r_2 et d . Il faut éliminer pour cela le cas dégénéré et supposer que les points A, B et C ne sont pas colinéaires, ce qui est équivalent à $x_2 - x_1 \neq 0$ et $x_4 \neq 0$.

On considère alors l'idéal $J = \langle f_1, \dots, f_6, 1 - (x_2 - x_1)s, 1 - x_4t \rangle \subseteq \mathbb{R}[d, s, t, r_1, r_2, x_1, \dots, x_7]$. L'idéal d'élimination $J \cap \mathbb{R}[d, s, t]$ contient le polynôme

$$g = (d^2 - r_1^2 + 2r_1r_2)(d^2 - r_1^2 - 2r_1r_2)$$

d'après le code Sage() ci-dessous

```
sage: R.<d,r1,r2,s,t,x1,x2,x3,x4,x5,x6> =
PolynomialRing(QQ,'d,r1,r2,s,t,x1,x2,x3,x4,x5,x6',order='invlex')
sage: f1 = (r2^2-x1^2)*x4-2*x1*r2*(x3-x1)
sage: f2 = (r2^2-x2^2)*x4-2*x2*r2*(x3-x2)
sage: f3 = 2*x5-x2-x1
sage: f4 = 2*x4*x6+2*x3*x5-2*x2*x5-x4^2-x3^2+x2^2
sage: f5 = d^2-(x6-r2)^2-x5^2
sage: f6 = r1^2-x6^2-(x5-x1)^2
sage: f7 = 1-(x2-x1)*s
sage: f8 = 1-x4*t
sage: I = (f1,f2,f3,f4,f5,f6,f7,f8)*R
```

```
sage: G = I.groebner_basis()
sage: G.parent()
Category of sequences in Multivariate Polynomial Ring in d, r1, r2,
s, t, x1, x2, x3, x4, x5, x6 over Rational Field
sage : len(G)
18
sage: G[17].factor()
(1/4) * (2*r1*r2 - r1^2 + d^2) * (2*r1*r2 + r1^2 - d^2)
```


Comme $d < r_1$, on en déduit l'équation

$$r_1^2 - 2r_1r_2 = d^2.$$

VIII.4.2. La formule de Héron.— La formule de Héron permet d'exprimer l'aire s d'un triangle quelconque en fonction des longueurs a , b et c de ses trois côtés :

$$s^2 = \frac{1}{16}(a+b+c)(a+b-c)(a-b+c)(-a+b+c).$$

Cette formule peut être obtenue par méthode d'élimination des indéterminées. Considérons le triangle suivant



Exercice 10. — Établir les relations suivantes :

$$b^2 = (a-x)^2 + y^2, \quad c^2 = x^2 + y^2, \quad 2s = ay.$$

Considérons l'idéal I de $\mathbb{R}[x, y, a, b, c, s]$ engendré par les équations précédentes :


$$I = \langle b^2 - (a-x)^2 - y^2, c^2 - x^2 - y^2, 2s - ay \rangle$$

Exercice 11. —

1. Décrire le deuxième idéal d'élimination $I \cap \mathbb{R}[a, b, c, s]$.
2. En déduire la formule de Héron.
3. On pose $p = (a+b+c)/2$ le demi périmètre du triangle. Vérifier que la formule de Héron est équivalente à

$$s^2 = p(p-a)(p-b)(p-c).$$

VIII.4.3. Formule de Brahmagupta.— La formule de Héron se généralise aux quadrilatères convexes inscriptibles, quadrilatères dont les sommets sont sur un même cercle.



Pour un tel quadrilatère dont les cotés sont de longueurs a , b , c et d , si l'on note $p = (a + b + c + d)/2$ le demi-périmètre du quadrilatère et s sa surface, on a alors la formule de Brahmagupta suivante :

$$s^2 = (p - a)(p - b)(p - c)(p - d).$$

(Si deux points sont confondus, on retrouve la formule de Héron.)

Exercice 12. —

1. En prenant pour origine le centre du cercle et pour coordonnées $A(x_1, x_2)$, $B(x_3, x_4)$, $C(x_5, x_6)$, $D(x_7, x_8)$, traduire les hypothèses par des équations algébriques. On vérifiera en particulier que l'aire "orientée" s satisfait l'équation suivante :

$$x_5x_8 - x_1x_8 - x_6x_7 + x_2x_7 + x_3x_6 - x_4x_5 + x_1x_4 - x_2x_3 - 2s = 0.$$

2. Soit I l'idéal correspondant à ces équations. En utilisant Sage, vérifier que l'idéal d'élimination $I \cap \mathbb{R}[a, b, c, d, s]$ est engendré par la polynôme $g = g_1g_2$ où

$$g_1 = 16s^2 + d^4 - 2c^2d^2 - 2b^2d^2 - 2a^2d^2 - 8abcd + c^4 - 2b^2c^2 - 2a^2c^2 + b^4 - 2a^2b^2 + a^4$$

et

$$g_2 = 16s^2 + d^4 - 2c^2d^2 - 2b^2d^2 - 2a^2d^2 + 8abcd + c^4 - 2b^2c^2 - 2a^2c^2 + b^4 - 2a^2b^2 + a^4.$$

3. Il suit que les hypothèses entraînent que $g_1 = 0$ ou $g_2 = 0$. Vérifier que $g_1 = 0$ correspond à la formule de Brahmagupta.